


EX 4.1: Escriu un programa en alt nivell que mostri una 'A' a la posició [4,8] de la pantalla, amb atribut invers, i una 'B' a la posició [4,9] amb atribut normal. Observa que la fila és la mateixa en els dos casos, així que sols cal escriure-la un cop.

```
.include "macros.s"
.include "crt0.s"
.text

main:
    $movei R3, 0x8000 ;posada en marxa
    $movei R1, 4
    out Rfil_pant, R1 ;fila 4
    $movei R1, 8
    out Rcol_pant, R1 ;col 8
    $movei R2, 0x100 + 'A
    out Rdat_pant, R2
    out Rcon_pant, R3


    addi R1,R1,1
    out Rcol_pant, R1 ;col 9
    $movei R2, 'B
    out Rdat_pant, R2 ;guardem el valor per imprimir-ho
    out Rcon_pant, R3 ;ho imprimim per pantalla
    halt
```



EX 4.2: Escriu un programa en alt nivell que esperi fins que es polsi una tecla qualsevol, i llavors escrigui una 'A' a la posició [4,8] de la pantalla, amb atribut normal.

```
.include "macros.s"
.include "crt0.s"
.text
    main:
        movi R0, 0
        out Rcon_tec, R0 ;per enquesta
    do:
        in R4, Rest_tec ;llegim si una tecla ha sigut presionada
        bz R4, do

        $movei R3, 0x8000 ;posada en marxa
        $movei R1, 4
        out Rfil_pant, R1 ;fila 4
        $movei R1, 8
        out Rcol_pant, R1 ;col 8
        $movei R2, 'A
        out Rdat_pant, R2 ;guardem el valor per imprimir-ho
        out Rcon_pant, R3 ;ho imprimim per pantalla
    halt
```



EX 4.3: Escriu una nova versió del programa en alt nivell anterior per tal que, en comptes d'una 'A', escrigui el caràcter associat a la tecla pulsada. Fixa't que es tracta de fer sols un petit canvi en la finalització. Recorda que per traduir el codi de rastreig del teclat es disposa del vector tteclat.

```
.include "macros.s"
.include "crt0.s"
.text


main:
    movi R0, 0
    out Rcon_tec, R0 ;per enquesta
    $movei R3, tteclat ; R3 = &tteclat[0]

do:
    in R1, Rest_tec ;llegim si una tecla ha sigut presionada
    bz R1, do

    in R1, Rdat_tec ;codi de rastreig
    add R1,R3,R1 ;R1=&tteclat[codi de rastreig]
    ldb R1, 0(R1) ;valor lletra en codi ASCII

    $movei R6, 0x8000 ;posada en marxa
    $movei R5, 4
    out Rfil_pant, R5 ;fila 4
    $movei R5, 8
    out Rcol_pant, R5 ;col 8
    out Rdat_pant, R1 ;guardem el valor per imprimir-ho
    out Rcon_pant, R6 ;ho imprimim per pantalla

halt
```



EX 4.4: Escriu en alt nivell una nova versió del programa anterior, tal que faci la mateixa tasca (llegir una tecla i escriure-la en pantalla) però repetidament, fins que la tecla polsada sigui una 'F'. Fixa't que es tracta sols d'insertar l'anterior programa dins un bucle.

```
.include "macros.s"
.include "crt0.s"
.text
    main:
        movi R0, 0
        out Rcon_tec, R0 ;per enquesta
        $movei R3, tteclat ; R3 = &tteclat[0]
        movi R0, 'F' ;R0<-CODI ASCII DE F

    do:
        in R1, Rest_tec ;llegim si una tecla ha sigut presionada
        bz R1, do

        in R1, Rdat_tec ;codi de rastreig
        add R1,R3,R1 ;R1=&tteclat[codi de rastreig]
        ldb R1, 0(R1) ;valor lletra en codi ASCII

        xor R4,R0,R1 ;si R1 = 'F' s'acaba el programa
        BZ R4, fi

        $movei R6, 0x8000 ;posada en marxa
        $movei R5, 4
        out Rfil_pant, R5 ;fila 4
        $movei R5, 8
        out Rcol_pant, R5 ;col 8
        out Rdat_pant, R1 ;guardem el valor per imprimir-ho
        out Rcon_pant, R6 ;ho imprimim per pantalla
        BNZ R4, do

    fi:
        halt
```

NO PODEM DEMOSTRAR-HO JA QUE HAURIAM DE FER UN VIDEO, PERO FUNCIONA

EX 4.5: Fes un programa en alt nivell que imprimeixi la paraula "Fi". Abans d'imprimir cada caràcter cal esperar que estigui preparada. I al final del programa, també.

```
.include "macros.s"
.include "crt0.s"
.text
    main:
    movi R0, 0
    out Rcon_imp, R0 ;per enquesta
    $movei R4, 0x8000 ;posem en marxa
    do1:
    in R5, Rest_imp ;encuesta
    bz R5, do1
    movi R1, 'F'
    out Rdat_imp, R1 ;imprimim F
    out Rcon_imp, R4 ;posar en marxa + enquesta

    do2:
    in R5, Rest_imp
    bz R5, do2
    movi R1, 'i'
    out Rdat_imp, R1 ;imprimim i
    out Rcon_imp, R4 ;posar en marxa + enquesta
    fi:
    in R5, Rest_imp
    bz R5, fi
    halt
```

```
#####
[2][ Printer: Ready ]#####Fi
#####
#####
#####
#####
#####
```

EX 4.6: Escriu en alt nivell una nova versió del programa anterior que escrigui el vector frase a la impressora. Fixa't que el vector és un string, per tant, acaba amb un byte a zero (valor binari 0). Es tracta doncs de fer un bucle que iteri fins a trobar el zero final.

```
.include "macros.s"
.include "crt0.s"
.data
    frase: .byte 'A','q','u','e','s','t',' ','p','r','o','g','r','a','m','a',' ','f','u','n','c','i','o','n','a',0;
.text
main:
    movi R0, 0
    $movei R2, frase ;@frase[0]
    out Rcon_imp, R0 ;per enquesta
    $movei R4, 0x8000 ;posem en marxa
do:
    in R5, Rest_imp ;encuesta
    bz R5, do
    ldb R1, 0(R2) ; R1 = frase[i]
    bz R1, fi ;final frase
    addi R2,R2,1 ; i=i+1
    out Rdat_imp, R1 ;imprimim frase[i]
    out Rcon_imp, R4 ;posar en marxa + enquesta
    bnz R1, do
fi:
    in R5, Rest_imp
    bz R5, fi
    halt
```

```
#####
[2][ Printer: Ready ]#####Aquest programa funciona
#####
#####
#####
#####
#####
```

EX 4.7 i 4.8:

1) Mostrar el valor de w (el codi del programa anterior).

2) Esperar fins que es polsi una tecla.

3) Modificar w segons quina sigui la tecla polsada:

- Si la tecla és una 'A', fer un desplaçament lògic de w (SHL), 1 lloc a l'esquerra.
- Si és una 'B', fer un desplaçament lògic de w (SHL), 1 lloc a la dreta.
- Si és una 'C', fer un desplaçament aritmètic de w (SHA), 1 lloc a la dreta.
- Si és una 'D', s'ha de dividir per dos l'enter w (DIV).
- Si és un número n (tecles '0'.... '9'), s'han de complementar els n bits de menys pes de w. Nota: aquesta operació es pot escriure en C així: $w = w^{((1 < n) - 1)}$; on l'operador ^ significa XOR bit a bit, i l'operador << significa desplaçament a Esquerra

```
.include "macros.s"
.include "crt0.s"
.data
    w: .word 0x4444
.text
    mostra:
        $MOVEI R1, 16          ;R1 = register int col
        $MOVEI R4, 0x8000      ;posada en marxa
        MOVI R2, 0
    out Rfil_pant, R2          ;fila 0
    $MOVEI R2, w
    LD R2, 0(R2)               ;R2 = w

    while:
        ADDI R1, R1, -1        ;col--
        OUT Rcol_pant, R1      ;columna pantalla = col
        MOVI R3, 1             ;R3 = 1
        AND R5, R3, R2         ;R5 menor bit de R2 = w
        $MOVEI R3, 0xFFFF     ;R3 = -1
        SHL R2, R2, R3         ;se mueve 1 bit a la derecha el valor R2
        OUT Rdat_pant, R5      ;se prepara R5 = menor bit de R2
        OUT Rcon_pant, R4      ;imprimir valor preparado
        MOVI R5, 0
        $CMPLT R5, R5, R1      ;si 0 < col, vuelve a while
        BNZ R5, while
        JMP R6

    main:
        bucle:
        $CAL R6, mostra        ;mostra w en binari
        $MOVEI R3, tteclat      ;R3 = &tteclat[0]
    do:
        IN R1, Rest_tec         ;llegim si una tecla ha sigut presionada
        BZ R1, do               ;R1 = 0 -> no tecla polsada
```

```

IN R1, Rdat_tec          ;codi de rastreig
ADD R1, R3, R1           ;R1 = &tteclat[codi de rastreig]
LDB R1, 0(R1)            ;R1 = lletra polsada
$MOVEI R3, w             ;R3 = &w
LD R4, 0(R3)             ;R4 = w

```

if1: ;desplaçament lògic de w (SHL), 1 lloc a l'esquerra.

```

$MOVEI R2, 'A
$CMPEQ R2,R1,R2
BZ R2, if2
MOVI R5, 1             ;R5 = 1
SHL R4, R4, R5         ;R4 = w desplazada 1 vez a la izquierda
ST 0(R3), R4 ;MEM[w] = R4
BNZ R2, bucle

```

if2: ;desplaçament lògic de w (SHL), 1 lloc a la dreta.

```

$MOVEI R2, 'B
$CMPEQ R2,R1,R2
BZ R2, if3
$MOVEI R5, 0xFFFF ;R5 = -1
SHL R4, R4, R5         ;R4 = w desplazada 1 vez a la derecha
ST 0(R3), R4 ;MEM[w] = R4
BNZ R2, bucle

```

if3: ;desplaçament aritmètic de w (SHA), 1 lloc a la dreta.

```

$MOVEI R2, 'C
$CMPEQ R2,R1,R2
BZ R2, if4
$MOVEI R5, 0xFFFF ;R5 = -1
SHA R4, R4, R5         ;R4 = w desplazada 1 vez a la derecha
ST 0(R3), R4 ;MEM[w] = R4
BNZ R2, bucle

```

if4: ;dividir per dos l'enter w (DIV).

```

$MOVEI R2, 'D
$CMPEQ R2,R1,R2
BZ R2, if5
$MOVEI R5, 2           ;R5 = 2
DIV R4, R4, R5         ;R4 = R4/2
ST 0(R3), R4 ;MEM[w] = R4
BNZ R2, bucle

```

if5:

```

$MOVEI R2, '0
$CMPGT R2,R1,R2 ;R2=1 -> R4>0
BZ R2, bucle ;R2 = 0, bucle
$MOVEI R2, '9
$CMPLT R2,R1,R2 ;R2=1 -> R4<=9
BZ R2, bucle ;R2 = 0, bucle

```



```

MOVI R0, 2          ;R0 fa la mascara
MOVI R5, '1'        ;R5=i=1
bucleaux:
    $CMPLAQ R6, R1, R5    ;i = teclapulsada -> R6=1
    BNZ R6, fibucle      ;R6=1, fibucle
    ADD R0, R0, R0        ;R0 mascara
    ADDI R5, R5, 1        ;R5++
    BZ R6, bucleaux       ;R6=0, bucleaux
fibucle:
    ADDI R0, R0, -1       ;R0= 001000 -> 000111
    XOR R4, R4, R0        ;w = complementar els n bits de
w
    ST 0(R3), R4          ;MEM[w] = R4
    BNZ R2, bucle
HALT

```

W:

```

#0100010001000100
#

```

A

```

#1000100010001000
#

```

C

```

#1100010001000100
#

```

B

```

#0110001000100010
#

```

D

```

#0011000100010001
#

```

5

```

#0011000100001110
#

```

EX 4.9: Comprova que quan w és negatiu ($w = 0xFFF9$), la divisió DIV per 2 dona un resultat diferent que el desplaçament SHA una posició a la dreta). Per què?

$W = 0xFFF9$:

```
#11111111111111001
#
```

D

```
#11111111111111101
#
```

C

```
#11111111111111100
#
```

En divisió si tenim $w=-7$, quan ho dividim entre 2 ens dona -3.5, llavors s'arrodoneix cap avall donant -4. Però en el sha, simplement desplaça els bits llavors si $w=-7$, fent el desplaçament donaria -3. Per això dona un resultat diferent fent la divisió que fent el desplaçament SHA.